A module to trade SFP entitlements between farm types and regions in CAPRI

Wolfgang Britz, Spring 2012

Background

With the so-called Mid Term Review of the Common Agricultural Policy, the so-called Single Farm Premium (SFP) as a decoupled payment was introduced which is implemented as a subsidy which does not require production, is subject to cross-compliance and paid per ha up to a number of entitlements. The original entitlements, defined on a hectare basis, had been distributed to farmers operating the land and not the land owners. Both land and entitlements can be traded independently from each other. After a sequence of reform steps, basically all crop production sectors are now included in the subsidy program, so that farmers can be assumed to have received entitlements for all hectares they cropped historically. The same was true from the beginning for the so-called regional implementation. If the land available to agriculture decreases, e.g. by urbanization, some entitlements cannot not longer be matched with a hectare of eligible land. Such unused entitlements are removed from the markets after a number of years.

In CAPRI, the assumption in the baseline is that all hectares used by agriculture are able to claim the SFP and that any unused entitlements had been removed so that the SFP becomes fully capitalized into land. Subsequent changes in the premiums including the SFP, prices or other policy instruments in a counterfactual run could decrease the marginal returns to agricultural land. Based on the land supply curve implemented in CAPRI, agricultural land use would shrink and some entitlements become unused. Vice versa, if changes let the marginal return to land increase, the entitlements become the limiting factor to claim the subsidy. The increase is thus mapped into an economic rent to the entitlement. If changes generate rents on entitlements in some farm types and not in others, one would assume that trade in entitlements will occur. A simple algorithm to trade the entitlement is now included in CAPRI and described below.

Implementation in the code

Switching on the entitlement

The trade module is implemented in the file "*policy\prem_entl_trade.gms*" which is included on demand in *capmod* and called in each iteration

```
*
$ifi %entl_trade%==on $include 'policy\prem_entl_trade.gms';
*
```

Policy files such as "*mtr_conv.gms*" can switch on the module:

```
$setglobal entl_trade on
```

The basic idea of the module is very simple: shift entitlements from farm type or regions which unused entitlements to other farm types or regions which have an economic rent on their entitlements. The trading entities should receive the very same premium on the entitlement for the current implementation in the code. One should hence set the trade level according to the regional level for which flat rate premiums are implemented as shown below in an example:

```
    Parameter p_premToDDTargetNuts(*) /
$ifi     %farm_m% == on    EU015000 2
$ifi not %farm_m% == on    EU015000 1
    /;
*
    p_premToDDTargetNuts(RMSSUP) = p_premToDDTargetNuts("EU015000");
    p_premToDDTargetNuts(MS) $ (SUM(PSDPAy_MTR_EL, p_premToDDTarget(MS,"%SIMY%",PSDPAY_MTR_EL,"DDSaps"))) = eps;

$setglobal entl_trade on

$ifi %farm_m% == on  $setglobal entl_trade_level NUTS2
$ifi %farm_m% == off $setglobal entl_trade_level NUTS1
```

How the entitlement trade works

The following code pieces are taken from “policy\prem_entl_trade.gms”. In a first step, the demand of entitlements is determined. The dual value does only provide an indication that entitlements are scarce, but not how many additional entitlements are needed. Accordingly, first, the average marginal value of the different type of entitlements is determined:

```
p_entlTrade(RU,"demand",step)
    =  sum( (PSDPAY_cutEndog,DDTarget) $ (     (not sameas(DDTarget,"DPGreen"))
                                                and (overShotEntl_.m(RU,PSDPAY_cutEndog,DDTarget) lt 0)),
            p_entlLimit(RU,PSDPAY_cutEndog,DDTarget,"Limit"));

p_entlTrade(RU,"value",step) $ p_entlTrade(RU,"demand",step)
    = -sum( (PSDPAY_cutEndog,DDTarget) $ (     (not sameas(DDTarget,"DPGreen"))
                                                and (overShotEntl_.m(RU,PSDPAY_cutEndog,DDTarget) lt 0)),
            p_entlLimit(RU,PSDPAY_cutEndog,DDTarget,"Limit") * overShotEntl_.m(RU,PSDPAY_cutEndog,DDTarget));

p_entlTrade(RU,"price",step) $ p_entlTrade(RU,"demand",step)
    = p_entlTrade(RU,"value",step)/p_entlTrade(RU,"demand",step);
```

From these a maximum of 10% is defined as the demand in each iteration

```
*
*    --- max 10 % of existing entitlements
*
  p_entlTrade(RU,"demand",step) =  p_entlTrade(RU,"demand",step) * 0.1;
```

In order to take differences in the marginal returns into account, an indicator based on the squared value is used:

```
  p_entlTrade(RU,"valueSqr",step) $ p_entlTrade(RU,"demand",step)
    = p_entlTrade(RU,"demand",step) * sqr(p_entlTrade(RU,"price",step));
```

It serves as the distribution key of unused entitlements, which are determined as follows:

```
*
* --- supply : 50% of unused entitlement
*
  p_entlTrade(RU,"supply",step)
      = sum( (PSDPAY_cutEndog,DDTarget) $ p_entlLimit(RU,PSDPAY_cutEndog,DDTarget,"Limit"),
            -MIN(0,v_sumEntl.l(RU,PSDPAY_cutEndog,DDTarget) - p_entlLimit(RU,PSDPAY_cutEndog,DDTarget,"Limit"))) * 0.50;
```

Next, the number of unused entitlements is stored:

```
*
* --- supply : 50% of unused entitlement
*
  p_entlTrade(RU,"supply",step)
      = sum( (PSDPAY_cutEndog,DDTarget) $ p_entlLimit(RU,PSDPAY_cutEndog,DDTarget,"Limit"),
            -MIN(0,v_sumEntl.l(RU,PSDPAY_cutEndog,DDTarget) - p_entlLimit(RU,PSDPAY_cutEndog,DDTarget,"Limit"))) * 0.50;
```

As seen, only 50% of the unused entitlements are released in any iteration. We next determine the size of the markets, i.e. total demand and supply:

```
*
* --- Aggregate demand,supply, value of farm types to NUTS2 level
*
  p_entlTrade(NUTS2Agg,"demand",step) =  sum( Types_to_r(NUTS2Agg,Types), p_entlTrade(Types,"demand",step));
  p_entlTrade(NUTS2Agg,"supply",step) =  sum( Types_to_r(NUTS2Agg,Types), p_entlTrade(Types,"supply",step));
  p_entlTrade(NUTS2Agg,"value",step)  =  sum( Types_to_r(NUTS2Agg,Types), p_entlTrade(Types,"value",step));
  p_entlTrade(NUTS2Agg,"valueSqr",step)  =  sum( Types_to_r(NUTS2Agg,Types), p_entlTrade(Types,"valueSqr",step));
```

The supply is then distributed according to the squared value of the individual demanders

```
p_entlLimit(Types,PSDPAY_cutEndog,DDTarget,"Limit") $ (    (overShotEntl_.m(Types,PSDPAY_cutEndog,DDTarget) lt 0)
                                                           and  p_entlLimit(Types,PSDPAY_cutEndog,DDTarget,"Limit"))
  = p_entlLimit(Types,PSDPAY_cutEndog,DDTarget,"Limit")
      + sum(types_to_r(Types,NUTS2Agg) $  p_entlTrade(NUTS2Agg,"value",step),
            p_entlTrade(Types,"valueSqr",step)
               * p_entlTrade(NUTS2Agg,"supply",step) / p_entlTrade(NUTS2Agg,"valueSqr",step));
```

An example printout

The following code snippet shows an example for a NUTS2 regions and the related farm types for a test run for Greece without the market module:

	S1	S2	S3	S4	S5	S99
EL110000.Demand	32.075	32.439	32.620	32.711	32.756	32.779
EL110000.Value	48766.234	24791.974	13600.485	7543.924	4360.840	2730.454
EL110000.SUPPLY	3.639	1.810	0.906	0.456	0.230	0.116
EL110000.valueSqr	788765.873	202072.288	59256.730	17969.995	5967.575	2332.077
EL110016.Demand	13.441	13.554	13.661	13.714	13.740	13.753
EL110016.Price	134.681	93.743	50.647	27.389	15.621	9.709
EL110016.Value	18102.989	12705.794	6918.624	3756.173	2146.387	1335.326
EL110016.valueSqr	243812.335	119107.489	35040.478	10287.771	3352.898	1296.493
EL110026.SUPPLY	2.300	1.144	0.573	0.288	0.146	0.073
EL110027.Demand	7.982	8.157	8.172	8.182	8.188	8.191
EL110027.Price	218.325	44.225	28.427	16.935	10.066	6.398
EL110027.Value	17426.013	3607.525	2322.885	1385.571	824.122	524.038
EL110027.valueSqr	380452.941	15954.221	6603.166	2346.486	829.521	335.274
EL110056.SUPPLY	0.510	0.253	0.127	0.064	0.032	0.016
EL110057.SUPPLY	0.327	0.163	0.082	0.041	0.021	0.010
EL110086.SUPPLY	0.502	0.250	0.125	0.063	0.032	0.016
EL110999.Demand	10.652	10.728	10.788	10.815	10.828	10.835
EL110999.Price	124.271	79.034	40.406	22.212	12.840	8.039
EL110999.Value	13237.233	8478.655	4358.976	2402.180	1390.331	871.090
EL110999.valueSqr	164500.597	67010.579	17613.085	5335.738	1785.156	700.309

As seen from above, we have two farm types in the starting situation which acts as demanders, i.e. have a marginal value on their entitlements (016 and 999). Their marginal value on the entitlement is quite high in the starting situation with > 125 € / entitlement. We have also a total of 3639 ha after the first round of unused entitlements which can be sold to the demanders. Distributing half of them (ca. 1800 ha) to the two demanders reduces the marginal value of the entitlements already below 95€, the next round distributed ca. 900 ha and brings the price down to 50€ until in the last round almost nothing is left for distribution and the value of the entitlements has dropped below 10€. The reader should note the trade is not yet taking into account in the income calculation of the farm types.

Finally, we come to the main point which motivated the introduction of that module. As indicated above, we interpret the SFP as a subsidy to agricultural land use which at the margin is capitalized in the land rent. It thus increases the marginal returns to land use in agriculture. In our baseline, we

start with a situation with an assumed equilibrium in land markets, i.e. marginal returns in agriculture including any subsidies are equal to marginal returns of alternative uses.

Reducing the SFP will render agricultural land use less competitive so that land owner will rent out less to agriculture and put the land into other uses. That effect can be clearly seen below in the first iteration: in the farm types where the SFP drops due to uniform SFP at NUTS2 in Greece, land use is reduced. Total land use in Greece drops by 1.2%. But if we re-distribute the subsidy between farm types, farms which were competing before with below average subsidies against alternative land use possibilities now would like to expand land use. Without additional entitlements, they cannot: the marginal return on the next ha drops by the SFP rate. But once they buy entitlements, they offset a larger part of the land loss: in step twp, the reduction is only about 0.6%. And towards the end, the basically a no-change in land use, as we would have assumed at the aggregated level if the same type of subsidy is paid on average with the same rate.



Summary

As the SFP entitlements are tradable, scenarios which lead to a more uniform distribution of premiums between farm types or regions run the risk to overestimate the effect on the land market. Without trade, farm types or regions with decreased premium rates will reduce land use and thus possess unused entitlements whereas farm types or regions increasing premiums rates cannot increase land use as they cannot claim the subsidy on additional hectares without additional entitlements. The small and rather simple module now available in CAPRI overcomes that problem by distributing unused entitlements to farm types or regions with an economic rent on the entitlements. The reader should however note that we do not yet take the effect on farm type or regional income from the trade into account.